




RDFS

Resource Description
Framework Schema



RDFS Concept vocabulary

1. Classify and organize resources

- **rdfs:Class** – the class of classes.
- **rdfs:subClassOf** – relate a class to a more general class (taxonomy).

2. Define and organize properties

- **rdf:Property** – the class of all properties.
- **rdfs:subPropertyOf** – relate a property to a more general property.
- **rdfs:domain** – declare the expected type of a property's subject.
- **rdfs:range** – declare the expected type of a property's object.

3. Link to datatypes

- **rdfs:Datatype** – the class of datatypes (e.g., `xsd:string`).

4. Provide human-readable documentation

- **rdfs:label** – a short human-readable name.
- **rdfs:comment** – a longer description.


5. Connect and document vocabularies

- **rdfs:seeAlso** – pointer to related information.
- **rdfs:isDefinedBy** – points to the defining RDF vocabulary/resource.

6. Top-level abstraction


- **rdfs:Resource** – everything describable by RDF is an instance of this.

SubClassOf



W₀

SubClassOf



\mathcal{W}_0

SubClassOf

$$(\mathbf{B} \subseteq \mathbf{A}) \wedge (x \in \mathbf{B}) \Rightarrow (x \in \mathbf{A})$$

\mathbf{B} rdfs:subClassOf \mathbf{A} .
?x rdf:type \mathbf{B} .
 \Rightarrow ?x rdf:type \mathbf{A} .

?b rdfs:subClassOf ?a .
?x rdf:type ?b .
 \Rightarrow ?x rdf:type ?a .

SubClassOf

`?b rdfs:subClassOf ?a .`
`?x rdf:type ?b .`
 \Rightarrow `?x rdf:type ?a .`

```
# QUERY 1: RDFS SUB-CLASS-OF REASONING
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
CONSTRUCT { ?x rdf:type ?superClass . }
WHERE
{
  ?class rdfs:subClassOf ?superClass .
  ?x rdf:type ?class .
}
```

SubClassOf

Result with reasoning enabled

```
# QUERY 2: ALL CLASS ASSIGNMENTS, BOTH ASSERTED AND INFERRED
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
CONSTRUCT { ?x a ?class . }
WHERE
{
  ?x a ?class .
  # FILTER(CONTAINS(STR(?class), "ens"))
}
```

SubPropertyOf

Reviewing our properties

```
# QUERY 3: REVIEW PROPERTIES
SELECT ?predicate (COUNT(?predicate) AS ?numberOf)
WHERE
{
    ?subject ?predicate ?object .
}
GROUP BY ?predicate
ORDER BY DESC(?numberOf)
```


SubPropertyOf

Getting triples related to a set of properties

```
# QUERY 4: GET TRIPLES RELATED TO A SET OF PROPERTIES
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX data: <http://www.webstep.no/workshop/data/>
PREFIX ens: <http://www.webstep.no/workshop/ens/>
CONSTRUCT
{
    ?equipment ?labelProp ?label .
}
WHERE
{
    VALUES ?labelProp {ens:code ens:tagNumber rdfs:label}
    ?equipment ?labelProp ?label .
}
```

SubPropertyOf

```
# QUERY 5: RDFS SUB-PROPERTY-OF REASONING
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
CONSTRUCT { ?x ?superProperty ?y . }
WHERE
{
    ?property rdfs:subPropertyOf ?superProperty .
    ?x ?property ?y .
}
```


SubPropertyOf

Reviewing the result with reasoning

```
# QUERY 6: REVIEWING THE RESULT WITH REASONING
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
CONSTRUCT { ?s ?p ?o . }
  WHERE {
    ?s ?p ?o .
    FILTER(?p = rdfs:label)
  }
```


Domain and Range

hasDiscipline




Domain and Range

hasDiscipline



Domain and Range

Inferring domain



Domain and Range

Inferring domain

```
# QUERY 7: CLASSIFICATION BASED ON DOMAIN
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
CONSTRUCT { ?subject a ?domain . }
WHERE {
  ?property rdfs:domain ?domain .
  ?subject ?property ?object .
}
```


Domain and Range

Validating domain infernal with reasoning enabled

```
# QUERY 8: REVIEW WITH REASONING ENABLED
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX ens:  <http://www.webstep.no/workshop/ens/>
CONSTRUCT { ?subject ?property ?object . }
WHERE {
    ?subject a ?inferredClass ;
             ?property ?object .
    FILTER(?inferredClass = ens:Equipment)
}
```


Domain and Range

Inferring range



Domain and Range

Inferring range

```
# QUERY 9: CLASSIFICATION BASED ON RANGE
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
CONSTRUCT { ?object a ?range . }
WHERE {
  ?property rdfs:range ?range .
  ?subject ?property ?object .
}
```

Domain and Range

Validate range infernal with reasoning enabled

```
# QUERY 10: REVIEW WITH REASONING ENABLED
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX ens: <http://www.webstep.no/workshop/ens/>
CONSTRUCT { ?subject ?property ?object . }
WHERE {
    ?subject a ?inferredClass ;
             ?property ?object .
    FILTER(?inferredClass = ens:Discipline)
}
```

Domain and Range

Globally enforced

Upload “4f Add discipline to system.ttl”

RDFS Concept vocabulary

1. Classify and organize resources

- **rdfs:Class** – the class of classes.
- **rdfs:subClassOf** – relate a class to a more general class (taxonomy).

2. Define and organize properties

- **rdf:Property** – the class of all properties.
- **rdfs:subPropertyOf** – relate a property to a more general property.
- **rdfs:domain** – declare the expected type of a property's subject.
- **rdfs:range** – declare the expected type of a property's object.

3. Link to datatypes

- **rdfs:Datatype** – the class of datatypes (e.g., `xsd:string`).

4. Provide human-readable documentation

- **rdfs:label** – a short human-readable name.
- **rdfs:comment** – a longer description.

5. Connect and document vocabularies

- **rdfs:seeAlso** – pointer to related information.
- **rdfs:isDefinedBy** – points to the defining RDF vocabulary/resource.

6. Top-level abstraction

- **rdfs:Resource** – everything describable by RDF is an instance of this.